NA12 Ortogonalni polinomi

Ivan Slapničar

12. studenog 2018.

1 Ortogonalni polinomi

Neka je

$$L(x_0,x_1,\ldots,x_n)$$

(pot)prostor razapet linearno nezavisnim vektorima (ili funkcijama) $x_0,x_1,\ldots,x_n$.

Radi se o skupu svih linearnih kombinacija zadanih vektora.

Koristeći *Gram-Schmidt-ov postupak ortogonalizacije* možemo izračunati *ortogonalnu bazu* tog (pot)prostora

$$y_0,y_1,\ldots,y_n,$$

za koju vrijedi

$$(y_i,y_j)=0,\quad i\neq j. \tag{1}$$

Neka je

$$\begin{aligned}
y_0&=x_0 && (1)\\
y_1&=x_1-\frac{(x_1,y_0)}{(y_0,y_0)}y_0 && (2)\\
y_2&=x_2-\frac{(x_2,y_0)}{(y_0,y_0)}y_0-\frac{(x_2,y_1)}{(y_1,y_1)}y_1 && (3)\\
&\vdots && (4)\\
y_n&=x_n-\sum_{j=0}^{n-1}\frac{(x_n,y_j)}{(y_j,y_j)}y_j. && (5)
\end{aligned}$$

Svaki y_j je linearna kombinacija od $x_0,x_1,\ldots,x_j$ pa su y_j linearno nezavisni i vrijedi

$$L(x_0,x_1,\ldots,x_n)=L(y_0,y_1,\ldots,y_n).$$

Direktnom provjerom se vidi da vrijedi (1).

Težinski skalarni produkt funkcija f i g na intervalu $[a,b]$ s težinom $\omega(x)>0$ je

$$(f,g)_\omega = \int_a^b f(x)g(x)\omega(x)\,dx$$

Funkcije f i g su *ortogonalne* ako je $(f,g)_\omega=0$.

Ortogonalni polinomi nastaju ortogonalizacijom polinoma

$$1,x,x^2,x^2,\ldots,x^n. \tag{2}$$

Različiti odabiri težinske funkcije daju različite sustave ortogonalnih polinoma.

1.1 Legendreovi polinomi

Ortogonalizirajmo sustav (2) uz

$$[a,b]=[-1,1],\quad \omega(x)=1,$$

koristeći paket `SymPy.jl` za simboličko računanje.

```
In [1]: using SymPy
```

```
In [2]: a=-1
        b=1
        n=8
        P=Array{Any,1}(missing,n)
        x=Sym("x")
        P[1]=x^0
        ω(x)=1
        for k=2:n
            P[k]=x^(k-1)
            for j=1:k-1
                P[k]=P[k]-integrate(x->x^(k-1)*P[j]*ω(x),a,b)*P[j]
                /integrate(x->P[j]*P[j]*ω(x),a,b)
            end
        end
```

Julia indeksiranje započima s 1 pa su svi indeksi pomaknuti, odnosno

$$P_0(x)=P[1],\ P_1(x)=P[2],\ldots$$

```
In [3]: P[1]
```

Out[3]:

$$1$$

In [4]: P[4]

Out[4]:

$$x^3 - \frac{3x}{5}$$

In [5]: P[6]

Out[5]:

$$x^5 - \frac{10x^3}{9} + \frac{5x}{21}$$

In [6]: P[7]

Out[6]:

$$x^6 - \frac{15x^4}{11} + \frac{5x^2}{11} - \frac{5}{231}$$

In [7]: P[8]

Out[7]:

$$x^7 - \frac{21x^5}{13} + \frac{105x^3}{143} - \frac{35x}{429}$$

Polinomi P_n su do na množenje konstantom jednaki *Legendreovim* polinomima

$$L_n(x) = \frac{1}{2^n n!}\frac{d^n}{dx^n}(x^2-1)^n, \quad n = 0,1,2,3,\ldots$$

In [8]:
```
n=8
L=Array{Any,1}(missing,n)
L[1]=x^0
for k=1:n-1
    L[k+1]=expand(diff((x^2-1)^k/(2^k*factorial(k)),x,k))
end
```

In [9]: `L[1], P[1]`

Out[9]: `(1, 1)`

In [10]:
```
L[2],P[2]
```

Out[10]:
```
(x, x)
```

In [11]:
```
L[4],P[4]
```

Out[11]:
```
(5*x^3/2 - 3*x/2, x^3 - 3*x/5)
```

In [12]:
```
L[7],P[7]
```

Out[12]:
```
(231*x^6/16 - 315*x^4/16 + 105*x^2/16 - 5/16,
 x^6 - 15*x^4/11 + 5*x^2/11 - 5/231)
```

In [13]:
```
P[7]
```

Out[13]:

$$x^6 - \frac{15x^4}{11} + \frac{5x^2}{11} - \frac{5}{231}$$

In [14]:
```
L[7]*16/231
```

Out[14]:

$$x^6 - \frac{15x^4}{11} + \frac{5x^2}{11} - \frac{5}{231}$$

Pored ortogonalnosti, vrijede sljedeća svojstva:

- $L_n(x)$ ima n različitih nul-točaka na intervalu $[-1,1]$,
- vrijedi *tročlana rekurzivna formula*:

$$L_{n+1}(x) = \frac{2n+1}{n+1}\,x\,L_n(x) - \frac{n}{n+1}L_{n-1}(x).$$

Izračunajmo polinome numerički i nacrtajmo ih:

In [15]:
```
using Polynomials
using Interact
```

In [16]:
```
n=40
L=Array{Any,1}(missing,n)
L[1]=Polynomials.Poly([1])
L[2]=Polynomials.Poly([0,1])
for i=3:n
    L[i]=(2*i-3)*L[2]*L[i-1]/(i-1)-(i-2)*L[i-2]/(i-1)
    # @show i, length(L[i])
end
```

```
In [17]: L[7]

Out[17]: Poly(-0.3125 + 6.5625*x^2 - 19.6875*x^4 + 14.4375*x^6)

In [20]: using PyPlot

In [21]: xx=range(-1,stop=1,length=300)
         @manipulate for k=1:n
             yy=polyval(L[k],xx)
             plot(xx,yy)
         end
```



1.2 Čebiševljevi polinomi

Čebiševljevi polinomi $T_n(x)$ nastaju ortogonalizacijom sustava (2) uz

$$[a,b]=[-1,1],\quad \omega(x)=\frac{1}{\sqrt{1-x^2}}.$$

Čebiševljevi polinomi imaju sljedeća svojstva:

- vrijedi

$$T_n(x)=\cos(n\arccos x),\quad n=0,1,2,3,\ldots,$$

- $T_n(x)$ ima n različitih nul-točaka na intervalu $[-1,1]$,

$$x_k = \cos\left(\frac{2k-1}{n}\frac{\pi}{2}\right), \quad k=1,\ldots,n,$$

- vrijedi *tročlana rekurzivna formula*:

$$\begin{aligned}
T_0(x) &= 1, \\
T_1(x) &= x, \\
T_{n+1}(x) &= 2\,x\,T_n(x) - T_{n-1}(x), \quad n=1,2,3,\ldots.
\end{aligned}$$

Napomena:
Rekurzivna formula slijedi iz *adicione formule*

$$\cos(n+1)\varphi + \cos(n-1)\varphi = 2\cos\varphi\cos n\varphi.$$

Ortogonalnost se dokazuje pomoću supstitucije

$$\arccos x = \varphi.$$

In [22]:
```
# Simbolički
n=8
T=Array{Any,1}(missing,n)
T[1]=x^0
T[2]=x
for k=2:n-1
    T[k+1]=expand(2*x*T[k]-T[k-1])
end
```

In [23]:
```
T[3]
```

Out[23]:

$$2x^2 - 1$$

In [24]:
```
T[7]
```

Out[24]:

$$32x^6 - 48x^4 + 18x^2 - 1$$

In [25]:
```
T[8]
```

Out[25]:

$$64x^7 - 112x^5 + 56x^3 - 7x$$

In [26]:
```
# Numerički
n=50
T=Array{Any,1}(missing,n)
T[1]=Polynomials.Poly([1])
T[2]=Polynomials.Poly([0,1])
for i=3:n
    T[i]=2*T[2]*T[i-1]-T[i-2]
    # @show i, length(T[i])
end
```

In [27]:
```
xx=range(-1,stop=1,length=300)
@manipulate for k=1:n
    yy=polyval(T[k],xx)
    plot(xx,yy)
end
```



1.3 Promjena intervala

Ortogonalni sustav funkcija Φ_i na intervalu $[-1,1]$ pomoću transformacije

$$\gamma : [a,b] \to [-1,1], \quad \gamma(x) = \frac{2x}{b-a} - \frac{a+b}{b-a}$$

prelazi u ortogonalni sustav funkcija na intervalu $[a, b]$

$$\Psi_i(x) = \Phi_i(\gamma(x)).$$

```
In [28]: a=1
         b=4
         xx=collect(range(a,stop=b,length=300))
         γ=2*xx/(b-a).-(b+a)/(b-a)
         @manipulate for k=1:n
             yy=polyval(T[k],γ)
             plot(xx,yy)
         end
```